

TUBO_GRADIENTE_TERMICO

```

finish
/clear

/filename,TERMICO
/title,TUBO SPESSO CON GRADIENTE TERMICO

! pi greco
PI = acos(-1)

! unità di misura: N,mm,s,K
! unità di massa: N*s^2/mm = 10^3*kg

! dimensioni
RI=100 !RAGGIO INTERNO
RE=200 !RAGGIO ESTERNO
L=100 !LUNGHEZZA ASSIALE MODELLO

! parametri meshing
ESZ = 2

! materiali
E1 = 206000 ! N/mm^2
PR1 = 0.3
DENS1 = 7.85e-9 ! 10^3*kg/mm^3
ALP1 = 1.43e-5 ! K^-1
K1 = 0.036 ! w/(mm*K)

! carichi
TAMB = 293 ! K
THOT = 403 ! K
TCOLD = 283 ! K

/PREP7

C***
C*** MODELLO SOLIDO
C***
RECTNG,RI,RE,0,L

! elementi e materiali
et,1,55,,,1
mp,kxx,1,K1

! carichi termici e vincoli
lsel,S,LOC,X,RI-0.01,RI+0.01
d1,ALL,,temp,THOT
lsel,S,LOC,X,RE-0.01,RE+0.01
d1,ALL,,temp,TAMB
alls

! meshing
aesize,all,ESZ
amesh,1

finish

/solu
! carichi e vincoli termici
lsel,S,LOC,X,RI-0.01,RI+0.01
d1,ALL,,temp,THOT
lsel,S,LOC,X,RE-0.01,RE+0.01
d1,ALL,,temp,TAMB
alls
solve
finish

```

TUBO_GRADIENTE_TERMICO

```

/post1
plnsol,temp
*ASK,IFL,RETURN PER CONTINUARE,0
plnsol,tf,x
*ASK,IFL,RETURN PER CONTINUARE,0

path,PHOT,2,,100
ppath,1,,RI,L/2
ppath,2,,RE,L/2
pdef,PHOT_AMB,temp
pasave
plpath,PHOT_AMB
*ASK,IFL,RETURN PER CONTINUARE,0
finish
/prep7
etchg,tts ! Element type
keyopt,1,3,1 !assialsimmetria
keyopt,1,1,3
mp,ex,1,E1
mp,prxy,1,PR1
mp,dens,1,DENS1
mp,reft,1,TAMB
mp,alpx,1,ALP1

/solu ! Re-enter the solution phase
ldread,temp,,,,,rth ! applica i carichi termici
tref,273
lsel,s,loc,y,0 ! vincoli allo spostamento assiale
dl,all,,UY,0
lsel,s,loc,y,L
dl,all,,UY,0
alls
dtran
/psbc,all,,1
eplot
solve
finish

/post1
paresu,PHOT
PDEF,SRAD,S,X
PDEF,SAXI,S,Y
PDEF,SCIR,S,Z,NOAV
PLPATH,SRAD,SCIR,SAXI

```